REP Prefix

REP repeats the following string instruction CX times. The use of CX is implied with the REP prefix. The decrement in CX doesn’t affect any flags and the jump is also independent of the flags, just like JCXZ.

REPE and REPNE Prefixes

REPE or REPZ repeat the following string instruction while the zero flag is set and REPNE or REPNZ repeat the following instruction while the zero flag is not set. REPE or REPNE are used with the SCAS or CMPS instructions. The other string instructions have nothing to do with the condition since they are performing no comparison. Also the initial state of flags before the string instruction does not affect the operation. The most complex operation of the string instruction is with these prefixes.

1.1. STOS Example – Clearing the Screen

We take the example of clearing the screen and observe that how simple and fast this operation is with the string instructions. Even if there are three instructions in a loop they have to be fetched and decoded with every iteration and the time of three instructions is multiplied by the number of iterations of the loop. In the case of string instructions, many operations are short circuited. The instruction is fetched and decoded once and only the execution is repeated CX times. That is why string instructions are so efficient in their operation. The program to clear the screen places 0720 on the 2000 words on the screen.

	; clear screen using string instructions

[org 0x0100]

              jmp  start

; subroutine to clear the screen

clrscr:       push es

              push ax

              push cx

              push di

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              xor  di, di             ; point di to top left column

              mov  ax, 0x0720         ; space char in normal attribute

              mov  cx, 2000           ; number of screen locations

              cld                     ; auto increment mode

              rep  stosw              ; clear the whole screen

              pop  di

              pop  cx

              pop  ax

              pop  es

              ret

start:        call clrscr             ; call clrscr subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	013
	A space efficient way to zero a 16bit register is to XOR it with itself. Remember that exclusive or results in a zero whenever the bits at the source and at the destination are same. This instruction takes just two bytes compared to “mov di, 0” which would take three. This is a standard way to zero a 16bit register.


Inside the debugger the operation of the string instruction can be monitored. The trace into command can be used to monitor every repetition of the string instruction. However screen will not be cleared inside the debugger as the debugger overwrites its display on the screen. CX decrements with every iteration, DI increments by 2. The first access is made at B800:0000 and the second at B800:0002 and so on. A complex and inefficient loop is replaced with a fast and simple instruction that does the same operation many times faster.

1.2. LODS Example – String Printing

The use of LODS with the REP prefix is not meaningful as only the last value loaded will remain in the register. It is normally used in a loop paired with a STOS instruction to do some block processing. We use LODS to pick the data, do the processing, and then use STOS to put it back or at some other place. For example in string printing, we will use LODS to read a character of the string, attach the attribute byte to it, and use STOS to write it on the video memory. 

The following example will print the string using string instructions.

	; hello world printing using string instructions

[org 0x0100]

              jmp  start

message:      db   'hello world'      ; string to be printed

length:       dw   11                 ; length of string

;;;;; COPY LINES 005-024 FROM EXAMPLE 7.1 (clrscr) ;;;;;

; subroutine to print a string 

; takes the x position, y position, attribute, address of string and 

; its length as parameters

printstr:     push bp

              mov  bp, sp

              push es

              push ax

              push cx

              push si

              push di

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  al, 80             ; load al with columns per row

              mul  byte [bp+10]       ; multiply with y position

              add  ax, [bp+12]        ; add x position

              shl  ax, 1              ; turn into byte offset

              mov  di,ax              ; point di to required location

              mov  si, [bp+6]         ; point si to string

              mov  cx, [bp+4]         ; load length of string in cx

              mov  ah, [bp+8]         ; load attribute in ah

              cld                     ; auto increment mode

nextchar:     lodsb                   ; load next char in al

              stosw                   ; print char/attribute pair

              loop nextchar           ; repeat for the whole string

              pop  di

              pop  si

              pop  cx

              pop  ax

              pop  es

              pop  bp

              ret  10

start:        call clrscr             ; call the clrscr subroutine

              mov  ax, 30        

              push ax                 ; push x position 

              mov  ax, 20

              push ax                 ; push y position 

              mov  ax, 1              ; blue on black attribute

              push ax                 ; push attribute 

              mov  ax, message

              push ax                 ; push address of message

              push word [length]      ; push message length

              call printstr           ; call the printstr subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	051
	Both operations are in auto increment mode.

	052-053
	DS is automatically initialized to our segment. ES points to video memory. SI points to the address of our string. DI points to the screen location. AH holds the attribute. Whenever we read a character from the string in AL, the attribute byte is implicitly attached and the pair is present in AX. The same effect could not be achieved with a REP prefix as the REP will repeat LODS and then start repeating STOS, but we need to alternate them.

	054
	CX holds the length of the string. Therefore LOOP repeats for each character of the string.


Inside the debugger we observe how LODS and STOS alternate and CX is only used by the LOOP instruction. In the original code there were four instructions inside the loop; now there are only two. This is how string instructions help in reducing code size. 

1.3. SCAS Example – String Length

Many higher level languages do not explicitly store string length; rather they use a null character, a character with an ASCII code of zero, to signal the end of a string. In assembly language programs, it is also easier to store a zero at the end of the string, instead of calculating the length of string, which is very difficult process for longer strings. So we delegate length calculation to the processor and modify our string printing subroutine to take a null terminated string and no length. We use SCASB with REPNE and a zero in AL to find a zero byte in the string. In CX we load the maximum possible size, which is 64K bytes. However actual strings will be much smaller. An important thing regarding SCAS and CMPS is that if they stop due to equality or inequality, the index registers have already incremented. Therefore when SCAS will stop DI would be pointing past the null character. 

	; hello world printing with a null terminated string

[org 0x0100]

              jmp  start

message:      db   'hello world', 0    ; null terminated string 

;;;;; COPY LINES 005-024 FROM EXAMPLE 7.1 (clrscr) ;;;;;

; subroutine to print a string 

; takes the x position, y position, attribute, and address of a null 

; terminated string as parameters

printstr:     push bp

              mov  bp, sp

              push es

              push ax

              push cx

              push si

              push di

              push ds 

              pop  es                 ; load ds in es

              mov  di, [bp+4]         ; point di to string

              mov  cx, 0xffff         ; load maximum number in cx

              xor  al, al             ; load a zero in al

              repne scasb             ; find zero in the string

              mov  ax, 0xffff         ; load maximum number in ax

              sub  ax, cx             ; find change in cx

              dec  ax                 ; exclude null from length

              jz   exit               ; no printing if string is empty

              mov  cx, ax             ; load string length in cx

              mov  ax, 0xb800

              mov  es, ax             ; point es to video base

              mov  al, 80             ; load al with columns per row

              mul  byte [bp+8]        ; multiply with y position

              add  ax, [bp+10]        ; add x position

              shl  ax, 1              ; turn into byte offset

              mov  di,ax              ; point di to required location

              mov  si, [bp+4]         ; point si to string

              mov  ah, [bp+6]         ; load attribute in ah

              cld                     ; auto increment mode

nextchar:     lodsb                   ; load next char in al

              stosw                   ; print char/attribute pair

              loop nextchar           ; repeat for the whole string

exit:         pop  di

              pop  si

              pop  cx

              pop  ax

              pop  es

              pop  bp

              ret  8

start:        call clrscr             ; call the clrscr subroutine

              mov  ax, 30        

              push ax                 ; push x position 

              mov  ax, 20

              push ax                 ; push y position 

              mov  ax, 1              ; blue on black attribute

              push ax                 ; push attribute 

              mov  ax, message

              push ax                 ; push address of message

              call printstr           ; call the printstr subroutine

              mov  ax, 0x4c00         ; terminate program

              int  0x21

	039-040
	Another way to load a segment register is to use a combination of push and pop. The processor doesn’t match pushes and pops. ES is equalized to DS in this pair of instructions.


Inside the debugger observe the working of the code for length calculation after SCASB has completed its operation.

